SOME EARTHQUAKE-MAGNITUDE STATISTICS

Author: J. Horner, jhorner@cybermesa.com
Last modified: 12 July 2019/1500 US Central Time

Platform: *Mathematica* ([1])

1.0 Introduction

This *Mathematica* ([1]) notebook computes some earthquake statistics based on worldwide sample of quakes with magnitude $\geqslant 5$. The results agree well with a television report about a prediction made by seismologists about the recent (~4/5 July 2019) Ridgecrest CA earthquake swarm.

2.0 Method and results

From the Wolfram Research database of earthquake data ([1]), retrieve the magnitudes from all records (one record per earthquake) of earthquakes with magnitude 5 or greater that occurred between 1 January 1990 and 20 September 2013 , inclusive, anywhere in the world. This retrieval requires online access to the database. The retrieval may time-out if the query-support functions of interest in the Wolfram database are not available at the time of the query. If a time-out occurs (the script will produce error messages if that happens), try running this script at a later time.

In[29]:=
```
magnitudes =
  #["Magnitude"] & /@ Values[EarthquakeData[All, 5, {{1990, 1, 1}, {2013, 9, 20}}]];
```

Count the number of earthquake records retrieved.

In[30]:=
```
numquakes = Length[magnitudes]
```

Out[30]= 36 115

Transpose the frequency/magnitude data, mapping it to 0.25-magnitude bins.

In[31]:=
```
transcoord = Transpose[{Range[5, 10, 0.25], BinCounts[magnitudes, {4.75, 10, 0.25}]}];
```

Plot log[frequency] vs. magnitude.

In[32]:=
```
ListLogPlot[transcoord, Joined → True]
```

Out[32]=



The plot strongly indicates the frequency/magnitude relationship is approximately a power law ([2], 5-6). Fit a power law to the (magnitude, frequency) relation.

In[33]:=
```
FindFit[transcoord, a x^(-b), {a, b}, x]
```

Out[33]=
```
{a → 8.40046 × 10^7, b → 5.66425}
```

The power characteristic (b) is > 3, which implies that both the mean ([2], 9) and the variance ([2], 9) are defined.

Compute the mean of the magnitudes.

In[34]:=
```
Mean[magnitudes]
```

Out[34]=
```
5.30947
```

Compute the standard deviation of the magnitudes.

In[35]:=
```
StandardDeviation[magnitudes]
```

Out[35]=
```
0.36034
```

Compute the probability that an earthquake has magnitude ⩾ 5.

In[36]:=
```
Probability[x ≥ 5, x ≃ EmpiricalDistribution[magnitudes]]
```

Out[36]=
```
1.
```

Compute the probability that an earthquake has magnitude ⩾ 6. The result agrees well with a report from television station KTLA : "(on 5 July 2019) seismologists had said there was a 6% chance of a magnitude 6 or larger earthquake striking" ([3]).

In[37]:=
```
Probability[x ≥ 6, x ≃ EmpiricalDistribution[magnitudes]]
```

Out[37]=
```
0.0641839
```

Compute the probability that an earthquake has magnitude ⩾ 7.

In[38]:=
```
Probability[x ≥ 7, x ≃ EmpiricalDistribution[magnitudes]]
```

Out[38]=
```
0.00373806
```

REFERENCES

[1] Wolfram Research. (2019). *Mathematica* v12.0 Home Edition. http://www.wolfram.com/mathematica-home-edition/. Accessed 11 July 2019.

[2] Newman MEJ. (2005). Power laws, Pareto distributions and Zipf's law. arXiv:cond_mat/0412004v2 [cond-mat.stat-mech] 9 Jan 2005. Accessed 11 July 2019.

[3] Martin E, Chambers R, and Knight N. (2019). Magnitude 7.1 earthquake follows July 4th Ridgecrest Temblor, becoming the 'mainshock' in a massive swarm. KTLA Channel 5. https://ktla.com/2019/07/05/-magnitude-5-0-earthquake-rattles-socal/.